Maple-assisted proof of formula for A295939

Robert Israel

30 November 2017

There are $2^8 = 256$ possible configurations for a 2×4 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is king-move adjacent to 1 or 3 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \\ b_5 & b_6 & b_7 & b_8 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6 b_7 b_8$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
     s:= floor((a-1)/16);
     if s <> (b-1) mod 16 then return 0 fi;
     s:= convert(s+16,base,2);
     r:= convert(16+floor((b-1)/16),base,2);
     t:= convert(16+ ((a-1) mod 16),base,2);
     M:= Vector(4);
     for i from 1 to 3 do if s[i] = 1 and s[i+1] = 1 then M[i]:= M
  [i]+1; M[i+1]:= 1 fi od;
     for i from 1 to 4 do if s[i]=1 then
       M[i]:= M[i]+r[i]+t[i];
       if i > 1 then M[i] := M[i]+r[i-1]+t[i-1] fi;
       if i < 4 then M[i] := M[i]+r[i+1]+t[i+1] fi;
       if M[i] <> 1 and M[i] <> 3 then return 0 fi;
     fi od;
     1
  end proc:
  T:= Matrix(256,256, q);
```

$$T := \begin{bmatrix} \textit{256 x 256 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \qquad \textbf{(1)}$$

Thus $a(n) = u\,T^n\,v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256):
  v:= Vector(256):
```

```
for i from 0 to 15 do u[16*i+1]:= 1; v[i+1]:= 1;
od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$4, 33, 120, 534, 2976, 13759, 65009, 325008, 1565481, 7539236 \quad (2)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{41} - t^{40} - 4\,t^{39} - 81\,t^{38} - 27\,t^{37} + 5\,t^{36} + 1630\,t^{35} + 952\,t^{34} + 891\,t^{33} - 13252\,t^{32} - 9662\,t^{31} - 12060\,t^{30} + 49580\,t^{29} + 47822\,t^{28} + 57446\,t^{27} - 73896\,t^{26} - 109840\,t^{25} - 119132\,t^{24} + 8355\,t^{23} + 106673\,t^{22} + 74595\,t^{21} + 57098\,t^{20} - 80052\,t^{19} + 461\,t^{18} - 35780\,t^{17} + 37583\,t^{16} + 4020\,t^{15} + 3892\,t^{14} - 10351\,t^{13} + 2104\,t^{12} - 4118\,t^{11} + 3246\,t^{10} - 2521\,t^{9} + 385\,t^{8} + 529\,t^{7} - 464\,t^{6} + 35\,t^{5} - 70\,t^{4} \quad (3)$$

This turns out to have degree 41. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{41} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 41, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 21, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^4\,(t^{16} + t^{15} + 3\,t^{14} - 18\,t^{13} - 8\,t^{12} - 12\,t^{11} + 34\,t^{10} + 56\,t^{9} + 16\,t^{8} + 3\,t^{7} - 38\,t^{6} - 23\,t^{5} - 5\,t^{4} + 8\,t^{3} + 4\,t^{2} - 5)\,(t^{21} - 2\,t^{20} - 5\,t^{19} - 52\,t^{18} + 12\,t^{17} + 55\,t^{16} + 505\,t^{15} + 34\,t^{14} + 70\,t^{13} - 1673\,t^{12} - 259\,t^{11} - 938\,t^{10} + 1632\,t^{9} - 663\,t^{8} + 899\,t^{7} - 985\,t^{6} + 542\,t^{5} - 19\,t^{4} - 89\,t^{3} + 104\,t^{2} - 7\,t + 14) \quad (4)$$

```
> Q:= unapply(t^21-2*t^20-5*t^19-52*t^18+12*t^17+55*t^16+505*
  t^15+34*t^14+70*t^13-1673*t^12-259*t^11-938*t^10+1632*t^9-663*
  t^8+899*t^7-985*t^6+542*t^5-19*t^4-89*t^3+104*t^2-7*t+14, t);
```

$$Q := t \to t^{21} - 2\,t^{20} - 5\,t^{19} - 52\,t^{18} + 12\,t^{17} + 55\,t^{16} + 505\,t^{15} + 34\,t^{14} + 70\,t^{13} - 1673\,t^{12} - 259\,t^{11} - 938\,t^{10} + 1632\,t^{9} - 663\,t^{8} + 899\,t^{7} - 985\,t^{6} + 542\,t^{5} - 19\,t^{4} - 89\,t^{3} + 104\,t^{2} - 7\,t + 14 \quad (5)$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 20.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^{16} + t^{15} + 3\,t^{14} - 18\,t^{13} - 8\,t^{12} - 12\,t^{11} + 34\,t^{10} + 56\,t^{9} + 16\,t^{8} + 3\,t^{7} - 38\,t^{6} - 23\,t^{5} - 5\,t^{4} + 8\,t^{3} + 4\,t^{2} - 5)\,t^4 \quad (6)$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-20 recurrence

$$\sum_{i=0}^{20} r_i\,b(i+n) = \sum_{i=0}^{20} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(19) = 0$. This would take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 19 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..19);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \tag{7}$$